Financial Modeling

Simon Benninga

second edition



Uses **EXCEL**



Financial Modeling

Financial Modeling, Second Edition
Simon Benninga

Financial Modeling

Simon Benninga

with a section on Visual Basic for Applications by Benjamin Czaczkes

SECOND EDITION

The MIT Press
Cambridge, Massachusetts London, England

Library of Congress Cataloging-in-Publication Data

Benninga, Simon,
Financial modeling / Simon Benninga; with a section on
Visual Basic for Applications by Benjamin Czaczkes.—2nd ed.
p. cm.
Includes bibliographical references and index.
ISBN 0-262-02482-9
1. Finance—Mathematical models. 2. Excel—Finance applications.
3. Microsoft Visual Basic for applications. I. Czaczkes, Benjamin.
II. Title.
HG173 .B46 2000
332.01'5118—dc21 00-035473

Dedication

To our parents: Helen and Noach Benninga, Esther and Alfred Czaczkes



Preface

Financial Modeling, Second Edition
Simon Benninga

Prev | Next

Preface

The purpose of this book remains to provide a "cookbook" for implementing common financial models in Excel. This edition has been expanded by six additional chapters, covering financial calculations, cost of capital, value at risk (VaR), real options, early exercise boundaries, and term-structure modeling. There is also an additional technical chapter containing a potpourri of Excel hints.

I am indebted to a number of people (in addition to those mentioned in the previous preface) for help and suggestions: Yoni Aziz, Michael Giacomo Bertolino, Michael J. Clarke, Beni Daniel, Hector Tassinari Eldridge, RazGilad, Doron Greenberg, Rick Labs, Allen Lee, Paul Legerer, Steve Rubin, Roger Shelor, Maja Sliwinski, Bob Taggart, Sandra van Balen, Ubbo Wiersema, and Khurshid Zaynutdinov. I also want to thank my editors, who again have been a great help: Nancy Lombardi, Peter Reinhart, Victoria Richardson, and Terry Vaughn.

As always I welcome suggestions and comments.

Simon Benninga
http://finance.wharton.upenn.edu/~benninga
<benninga@post.tau.ac.il>

Top
Prev | Next



Preface to the First Edition

Financial Modeling, Second Edition

Simon Benninga

Prev | Next

Preface to the First Edition

Like its predecessor *Numerical Techniques in Finance*, this book presents some important financial models and shows how they can be solved numerically and/or simulated using Excel. In this sense this is a finance "cookbook"; like any cookbook, it gives recipes with a list of ingredients and instructions for making and baking. As any cook knows, a recipe is just a starting point; having followed the recipe a number of times, you can think of your own variations and make the results suit your tastes and needs.

Financial Modeling covers standard financial models in the areas of corporate finance, financial statement simulation, portfolio problems, options, portfolio insurance, duration, and immunization. Clear and concise explanations are provided in each case for the implementation of the models using Excel. Very little theory is offered except where necessary to understand the numerical implementations.

While Excel is often inappropriate for high-level, industrial-strength calculations (portfolios are an example), it is an excellent tool for understanding the computational intricacies involved in financial modeling. It is often the case that the fullest understanding of the models comes by calculating them, and Excel is one of the most accessible and powerful tools available for this purpose.

Along the way a lot of students, colleagues, and friends (these are nonexclusive categories) have helped me with advice and comments. In particular I would like to thank Olivier Blechner, Miryam Brand, Elizabeth Caulk, John Caulk, Benjamin Czaczkes, John Ferrari, John P. Flagler, Kunihiko Higashi, Julia Hynes, Don Keim, Anthony Kim, Ken Kunimoto, Philippe Nore, Nir Sharabi, Mark Thaler, Terry Vaughn, and Xiaoge Zhou.

Finally, my thanks go to a wonderful set of editors: Nancy Lombardi, Peter Reinhart, Victoria Richardson, and Terry Vaughn.

Top

Prev | Next



Part I - Corporate Finance Models

Financial Modeling, Second Edition

Simon Benninga

Part I: Corporate Finance Models

Chapter List

The six chapters that open *Financial Modeling* cover some problems in corporate finance that are highly numerically intensive. Chapters 1 and 2 are a review of some finance basics. Chapter 1 is an introduction to basic financial calculations using Excel. Almost all of the applications discussed center on variations of the discounted-cash-flow method. The cost of capital, discussed in Chapter 2, is the rate at which corporate cash flows are discounted to arrive at enterprise value. Calculating this rate is not trivial and involves a combination of some theoretical models and numerical computation.

Chapter 3 shows how to build pro forma models, which simulate the corporate income statement and balance sheets. Pro forma models are at the heart of many corporate finance applications, including business plans, credit analyses, and valuations. The models require a mixture of finance, accounting, and Excel. In Chapter 4 we use pro forma models to do a valuation of a firm; the simple example we develop is typical of an exercise that accompanies many merger and acquisition valuations.

Chapters 5 and 6 discuss the financial analysis of leasing. In Chapter 5 we concentrate on the basic lease/purchase decision using the equivalent-loan method. An appendix to Chapter 5 discusses some tax and accounting considerations relating to leases. Chapter 6 discusses the financial analysis of leveraged lease arrangements, including a discussion of the multiple-phases method of Statement 13 of the Financial Accounting Standards Board (FASB 13). The multiple-phases-method rate of return is a hybrid internal rate of return (IRR), and Excel can easily be used to calculate this return.



Chapter 1: Basic Financial Calculations

1.1 Introduction

This chapter aims to give you some finance basics and their Excel implementation. If you have had a good introductory course in finance, most of the topics will probably be superfluous.

This chapter covers the following:

- Net present value (NPV)
- Internal rate of return (IRR)
- Future value
- Pension and accumulation problems
- Continuously compounded interest

Almost all financial problems center on finding the *value today* of a series of *cash receipts over time.* The cash receipts (or cash flows, as we will call them) may be certain or uncertain. In this chapter we analyze the values of nonrisky cash flows—future receipts that we will receive with absolute certainty.

The basic concept to which we will return over and over is the concept of *opportunity cost.* Opportunity cost is the return that would be required of an investment to make it a viable alternative to other, similar, investments.[1] As illustrated in this chapter, when we calculate the net present value, we use the investment's opportunity cost as a discount rate. When we calculate the internal rate of return, we compare the calculated return to the investment's opportunity cost to judge its value.

[1]In the financial literature you will find many synonyms for *opportunity cost,* among them *discount rate, cost of capital,* and *interest rate.* When it is applied to risky cash flows (as in the next chapter), we will sometimes call the opportunity cost the risk-adjusted discount rate (RADR) or the weighted average cost of capital (WACC).



Chapter 1 - Basic Financial Calculations

Financial Modeling, Second Edition

Simon Benninga

Prev | Next

1.2 Present Value (PV) and Net Present Value (NPV)

Both concepts, present value and net present value, are related to the value *today* of a set of future anticipated cash flows. As an example, suppose we are valuing an investment that promises $100 per year at the end of this and the next four years. We suppose that there is no doubt that this series of five payments of $100 each will actually be paid. If a bank would pay us an annual interest rate of 10 percent on a five-year deposit, then this 10 percent is the investment's opportunity cost, the alternative benchmark return to which we want to compare the investment. We may calculate the value of the investment by discounting its cash flows using this opportunity cost as a discount rate:

	A	B	C	D
2	Discount rate	10%		
3	Present value	$379.08	<-- =NPV(B2,B7:B11)	
4				
5		Cash		
6	Year	flow		
7	1	100		
8	2	100		
9	3	100		
10	4	100		
11	5	100		

The *present value* (PV) of $379.08 is the value today of the investment.

Suppose this investment was being sold for $400. Clearly it would not be worth its purchase price, since—given the alternative return (discount rate) of 10 percent—the investment is worth only $379.08. The *net present value* (NPV) is the applicable concept here. Denoting by r the discount rate applicable to the investment, the NPV is calculated as follows:

$$NPV = CF_0 + \sum_{t=1}^{N} \frac{CF_t}{(1+r)^t}$$

where CF_t is the investment's cash flow at time t and CF_0 is today's cash flow:

	F	G	H	I	J
2	Discount rate	10%			
3	Net present value	-20.92	<-- =G7+NPV(G2,G8:G12)		
4					
5		Cash			
6	Year	flow			
7	0	-400			
8	1	100			
9	2	100			
10	3	100			
11	4	100			
12	5	100			

A Note about Nomenclature

Excel's language about discounted cash flows differs somewhat from the standard finance nomenclature. Excel uses the letters NPV to denote the present value (*not* the net present value) of a series of cash flows.

To calculate the finance *net present value* of a series of cash flows using Excel, we have to calculate the *present value* of the future cash flows (using the Excel **NPV** function) and subtract from this present value the time-zero cash flow. (This is often the cost of the asset in question.)



Chapter 1 - Basic Financial Calculations

Financial Modeling, Second Edition

Simon Benninga

Prev | Next

1.3 The Internal Rate of Return (IRR) and Loan Tables

We continue with the same example. Suppose that we indeed paid $400.00 for this series of cash flows. The *internal rate of return* (IRR) is defined as the compound rate of return r that makes the NPV equal to zero:

$$CF_0 + \sum_{t=1}^{N} \frac{CF_t}{(1+r)^t} = 0$$

Excel's function **IRR** will solve this problem; note that the IRR includes as arguments *all* of the cash flows of the investment, including the first (in this case negative) cash flow of −400:

	A	B	C	D
15	IRR	7.931%	<-- =IRR(B19:B24)	
16	NPV	-20.92		
17		Cash		
18	Year	flow		
19	0	-400		
20	1	100		
21	2	100		
22	3	100		
23	4	100		
24	5	100		

The IRR is the compound *rate of return paid by the investment*. To understand this point fully, it helps to make the following table:

	E	F	G	H	I	J	K
15	LOAN TABLE			Division of payment			
16				between interest			
17	=-B19	Principal	Payment	and return of principal			
18		at beginning	at end				
19	year	of year	of year	Interest	Principal		
20	1	400.00	100	31.72	68.28	=G20-H20	
21	2	331.72	100	26.31	73.69		
22	3	258.03	100	20.46	79.54		
23	4	178.50	100	14.16	85.84		
24	5	92.65	100	7.35	92.65		
25							
26	=F20-I20		=$B$15*F20				
27							

The *loan table* divides each of the payments made by the asset into an interest component and a return-of-principal component. The interest component at the end of each year is the IRR times the principal balance at the beginning of that year. Notice that the principal at the beginning of the last year ($92.65 in the example) exactly equals the return of principal at the end of that year.

We can actually use the loan table to find the internal rate of return. Consider an investment costing $1,000 today that pays off at the end of years 1, 2, …, 5.

As the following loan table shows, the IRR of this investment is larger than 15 percent:

	A	B	C	D	E	F	G
2	Cost	1000					
3	IRR?	15.00%					
4							
5							
6	LOAN TABLE			Division of payment			
7				between interest			
8	=-B2	Principal	Payment	and return of principal			
9		at beginning	at end				
10	Year	of year	of year	Interest	Principal		
11	1	1000.00	300	150.00	150.00	=C11-D11	
12	2	850.00	200	127.50	72.50		
13	3	777.50	150	116.63	33.38		
14	4	744.13	600	111.62	488.38		
15	5	255.74	900	38.36	861.64		
16	6	-605.89					
17	=B11-E11		=$B$3*B11				
18							

Note that we have added an extra cell (B16) to this example. If the interest rate in cell B3 is indeed the IRR, then cell B16 should be 0. We can now use Excel's **Goal Seek** (found on the **Tools** menu) to calculate the IRR:

	A	B	C	D	E	F	G	H	I
1									
2	Cost	1000							
3	IRR?	15.00%							
4									
5									
6	LOAN TABLE			Division of payment					
7				between interest					
8	=-B2	Principal	Payment	and return of principal					
9		at beginning	at end						
10	Year	of year	of year	Interest	Principal				
11	1	1000.00	300	150.00	150.00	=C11-D11			
12	2	850.00	200	127.50	72.50				
13	3	777.50	150	116.63	33.38				
14	4	744.13	600	111.62	488.38				
15	5	255.74	900	38.36	861.64				
16	6	-605.89							
17	=B11-E11		=$B$3*B11						
18									

Goal Seek
Set cell: $B$16
To value: 0
By changing cell: $B$3
OK Cancel

You can see the result in the following display:

	A	B	C	D	E	F	G
2	Cost	1000					
3	IRR?	24.44%					
4							
5							
6	LOAN TABLE			Division of payment			
7				between interest			
8	=-B2	Principal	Payment	and return of principal			
9		at beginning	at end				
10	Year	of year	of year	Interest	Principal		
11	1	1000.00	300	244.36	55.64	=C11-D11	
12	2	944.36	200	230.76	-30.76		
13	3	975.13	150	238.28	-88.28		
14	4	1063.41	600	259.86	340.14		
15	5	723.26	900	176.74	723.26		
16	6	0.00					
17	=B11-E11		=$B$3*B11				
18							

Of course, we could have simplified life by just using the **IRR** function:

	A	B	C	D
22	**Year**	**Cash flow**		
23	0	-1000		
24	1	300		
25	2	200		
26	3	150		
27	4	600		
28	5	900		
29				
30	IRR	24.44%	<-- =IRR(B23:B28)	



Chapter 1 - Basic Financial Calculations

Financial Modeling, Second Edition

Simon Benninga

1.4 Multiple Internal Rates of Return

Sometimes a series of cash flows has more than one IRR. In the next example we can tell that the cash flows in cells B35:B40 have two IRRs, since the NPV graph crosses the *x*-axis twice.

	A	B	C	D	E	F	G	H	I
1	**MULTIPLE INTERNAL RATES OF RETURN**								
2									
3	Discount rate	6%							
4	NPV	-3.99	<-- =NPV(B3,B9:B13)+B8				**DATA TABLE**		
5							**Discount**		
6		Cash					**rate**	**NPV**	
7	Year	flow						-3.99	
8	0	-145					0%	-20.00	
9	1	100					3%	-10.51	
10	2	100					6%	-3.99	
11	3	100					9%	0.24	
12	4	100					12%	2.69	
13	5	-275					15%	3.77	
14							18%	3.80	
15							21%	3.02	
16							24%	1.62	
17							27%	-0.24	
18							30%	-2.44	
19							33%	-4.90	
20							36%	-7.53	
21							39%	-10.27	
22									
23							**Note:** For a discussion of how		
24							to create data tables in Excel		
25							see Chapter 26.		
26									
27									
28									
29									
30									
31	**Identifying the two IRRs**								
32	First IRR	8.78%	<-- =IRR(B8:B13,0)						
33	Second IRR	26.65%	<-- =IRR(B8:B13,0.3)						



Excel's **IRR** function allows us to add an extra argument that will help us find both IRRs. Instead of writing IRR (B8:B13), we write IRR(B8:B13,guess). The argument **guess** is a starting point for the algorithm which Excel uses to find the IRR; by adjusting the **guess,** we can identify both the IRRs. Cells B59 and B60 give an illustration.

There are two things we should note about this procedure.

1. The argument **guess** merely has to be close to the IRR; it is not unique. For example by setting the guesses equal to 0.1 and 0.5, we will still get the same IRRs:

	A	B	C	D
31	**Identifying the two IRRs**			
32	First IRR	8.78%	<-- =IRR(B8:B13,0.1)	
33	Second IRR	26.65%	<-- =IRR(B8:B13,0.5)	
34				

2. In order to identify the number and the approximate value of the IRRs, it helps greatly to graph the NPV of the investment as a function of various discount rates (as we have already done). The internal rates of return are then the points where the graph crosses the *x*-axis, and the approximate location of these points should be used as the guesses in the IRR function.[2]

From a purely technical point of view, a set of cash flows can have multiple IRRs only if it has at least two changes of sign. Many "typical" cash flows have only one change of sign. Consider, for example, the cash flows from purchasing a bond having a 10 percent coupon, a face value of $1,000, and eight more years to maturity. If the current market price of the bond is $800, then the stream of cash flows changes signs only once (from negative in year 0 to positive in years 1–8). Thus there is only one IRR:



	A	B	C	D	E	F	G
1	BOND CASH FLOWS						
2							
3	Year	Cash flow				Data table:	Effect of
4	0	-800				discount rate on NPV	
5	1	100					1000.00
6	2	100				0%	1000.00
7	3	100				2%	786.04
8	4	100				4%	603.96
9	5	100				6%	448.39
10	6	100				8%	314.93
11	7	100				10%	200.00
12	8	1100				12%	100.65
13						14%	14.45
14	IRR	14.36%	<-- =IRR(B4:B12)			16%	-60.82
15						18%	-126.21
16						20%	-183.72
17							
18							

[2]If you don't put in a guess (as we did in the previous section), Excel defaults to a guess of 0. Thus, in the current example, IRR(B8:B13) will return 8.78 percent.

Chapter 1 - Basic Financial Calculations

Financial Modeling, Second Edition

Simon Benninga

1.5 Flat Payment Schedules

Another problem: You take a loan for $10,000 at an interest rate of 7 percent per year. The bank wants you to make a series of payments that will pay off the loan and the interest over six years. We can use Excel's **PMT** function to determine how much should each annual payment be:



Notice that we have put "PV"—Excel's nomenclature for the initial loan principal—with a minus sign. Otherwise Excel returns a negative payment (a minor irritant).

You can confirm that this answer is correct by creating a loan table:

	A	B	C	D	E	F	G	H
1			FLAT PAYMENT SCHEDULES					
2								
3	Loan principal	10,000						
4	Interest rate	7%						
5	Loan term	6	<-- Number of years over which loan is repaid					
6	Annual payment	2,097.96	<-- To be made at end of each year					
7								
8			Principal at	Payment	Split payment into:			
9		Year	beginning	at end of		Return of	=$B$4*C11	
10			of year	year	Interest	principal		
11		1	10,000.00	2,097.96	700.00	1,397.96		
12		2	8,602.04	2,097.96	602.14	1,495.82	=D11-E11	
13	=C11-F11	3	7,106.23	2,097.96	497.44	1,600.52		
14		4	5,505.70	2,097.96	385.40	1,712.56		
15		5	3,793.15	2,097.96	265.52	1,832.44		
16		6	1,960.71	2,097.96	137.25	1,960.71		



Chapter 1 - Basic Financial Calculations

Financial Modeling, Second Edition

Simon Benninga

1.6 Future Values and Applications

We start with a triviality. Suppose you deposit $1,000 in an account, leaving it there for 10 years. Suppose the account draws annual interest of 10 percent. How much will you have at the end of 10 years? The answer, as shown in the following spreadsheet, is $2,593.74.

	A	B	C	D	E	F
1		SIMPLE FUTURE VALUE				
2						
3	Interest	10%				
4						
5	Year	Account	Interest	Total in		
6		balance	earned	account		
7		beg. year	during year	end of year		
8	0	1,000.00	100.00	1,100.00	<-- =C8+B8	
9	1	1,100.00	110.00	1,210.00		
10	2	1,210.00	121.00	1,331.00	=$B$3*B8	
11	3	1,331.00	133.10	1,464.10		
12	4	1,464.10	146.41	1,610.51		
13	5	1,610.51	161.05	1,771.56		
14	6	1,771.56	177.16	1,948.72		
15	7	1,948.72	194.87	2,143.59		
16	8	2,143.59	214.36	2,357.95		
17	9	2,357.95	235.79	2,593.74		
18	10	2,593.74				
19				=D8		
20						
21	A simpler way		2593.74	<-- =1000*(1+B3)^10		

As cell C21 shows, you don't need all these complicated calculations: The *future value* of $1,000 in 10 years at 10 percent per year is given by

$$FV = 1{,}000 * (1 + 10\%)^{10} = 2{,}593.74$$

Now consider the following, slightly more complicated, problem: Again, you intend to open a savings account. Your initial deposit of $1,000 this year will be followed by a similar deposit at the beginning of years 1,2, …, 9. If the account earns 10 percent per year, how much will you have in the account at the start of year 10?

This problem is easily modeled in Excel:

	A	B	C	D	E	F	G
1		FUTURE VALUE WITH ANNUAL DEPOSITS					
2							
3	Interest	10%					
4							
5	Year	Account	Deposit at	Interest	Total in		
6		balance	beginning	earned	account		
7		beg. year	of year	during year	end of year		
8	0	0.00	1,000	100.00	1,100.00	<-- =D8+C8+B8	
9	1	1,100.00	1,000	210.00	2,310.00		
10	2	2,310.00	1,000	331.00	3,641.00	=$B$3*(C8+B8)	
11	3	3,641.00	1,000	464.10	5,105.10		
12	4	5,105.10	1,000	610.51	6,715.61		
13	5	6,715.61	1,000	771.56	8,487.17		
14	6	8,487.17	1,000	948.72	10,435.89		
15	7	10,435.89	1,000	1,143.59	12,579.48		
16	8	12,579.48	1,000	1,357.95	14,937.42		
17	9	14,937.42	1,000	1,593.74	17,531.17		
18	10	17,531.17					
19			=E8				
20							
21		Future value		$17,531.17	<-- =FV(B3,A18,-1000,,1)		

Thus the answer is that we will have $17,531.17 in the account at the beginning of year 10. This same answer can be represented as a formula that sums the future values of each deposit.

$$\text{Total at beginning of year } 10 = 1{,}000 * (1 + 10\%)^{10} + 1{,}000 * (1 + 10\%)^{9} + \cdots + 1{,}000 * (1 + 10\%)^{1} = \sum_{t=1}^{10} 1{,}000 * (1 + 10\%)^{t}$$

An Excel Function Note from cell D21 that Excel has a function **FV** that gives this sum. The dialog box brought up by **FV** is the following:



We note three things about this function:

1. For positive deposits FV returns a negative number. There is an explanation for why this function is programmed in this way, but basically this outcome is an irritant. To avoid negative numbers, we have put the **Pmt** in as −1,000.
2. The line **Pv** in the dialog box refers to a situation where the account has some initial value other than 0 when the series of deposits is made. In this example, this line has been left blank, indicating that the initial account value is zero.
3. As noted in the picture, "Type" (either 1 or 0) refers to whether the deposit is made at the beginning or the end of each period.



Chapter 1 - Basic Financial Calculations

Financial Modeling, Second Edition

Simon Benninga

1.7 A Pension Problem—Complicating the Future Value Problem

A typical exercise follows. You are 55 years old and intend to retire at age 60. To make your retirement easier, you intend to start a retirement account.

- At the beginning of each of years 0, 1, 2, …, 4 (i.e., starting today and for each of the next four years), you intend to make a deposit into the retirement account. You think that the account will earn 8 percent per year.
- After retirement at age 60, you anticipate living eight more years.[3] During each of these years you want to withdraw $30,000 from your retirement account. Of course, account balances will continue to earn 8 percent.

How much should you deposit annually in the account? The following spreadsheet fragment shows how easily you can go wrong in this kind of problem—in this case, you've calculated that in order to provide $30,000 per year for eight years, you need to contribute $240,000/5 = $48,000 in each of the first five years. As the spreadsheet shows, you'll end up with a lot of money at the end of eight years! (The reason—you've ignored the powerful effects of compound interest. If you set the interest rate in the spreadsheet equal to 0 percent, you'll see that you're right.)

	A	B	C	D	E	F	G	H
1			A RETIREMENT PROBLEM					
2								
3	Interest	8%						
4	Annual deposit	48,000						
5	Annual retirement withdrawal	30,000						
6							=$B$3*(D10+C10)	
7		Year	Account	Deposit	Interest	Total in		
8			balance	at beginning	earned	account		
9			beg. year	of year	during year	end of year		
10		0	0.00	48,000	3,840.00	51,840.00	<-- =E10+D10+C10	
11		1	51,840.00	48,000	7,987.20	107,827.20		
12		2	107,827.20	48,000	12,466.18	168,293.38		
13		3	168,293.38	48,000	17,303.47	233,596.85		
14		4	233,596.85	48,000	22,527.75	304,124.59		
15		5	304,124.59	-30,000	21,929.97	296,054.56		
16		6	296,054.56	-30,000	21,284.36	287,338.93		
17		7	287,338.93	-30,000	20,587.11	277,926.04		
18		8	277,926.04	-30,000	19,834.08	267,760.12		
19		9	267,760.12	-30,000	19,020.81	256,780.93		
20		10	256,780.93	-30,000	18,142.47	244,923.41		
21		11	244,923.41	-30,000	17,193.87	232,117.28		
22		12	232,117.28	-30,000	16,169.38	218,286.66		

There are two ways to solve this problem. The first involves Excel's Solver. This can be found on the Tools menu.[4]



Clicking on the **Solver** makes a dialog box appear; here we've filled it in:



If we now click on the **Solve** box, we get the answer:



	A	B	C	D	E	F	G	H
1			A RETIREMENT PROBLEM					
2								
3	Interest	8%						
4	Annual deposit	29,387						
5	Annual retirement withdrawal	30,000						
6							=$B$3*(D10+C10)	
7		Year	Account	Deposit	Interest	Total in		
8			balance	at beginning	earned	account		
9			beg. year	of year	during year	end of year		
10		0	0.00	29,387	2,350.92	31,737.48	<-- =E10+D10+C10	
11		1	31,737.48	29,387	4,889.92	66,013.95		
12		2	66,013.95	29,387	7,632.04	103,032.54		
13		3	103,032.54	29,387	10,593.53	143,012.62		
14		4	143,012.62	29,387	13,791.93	186,191.10		
15		5	186,191.10	-30,000	12,495.29	168,686.39		
16		6	168,686.39	-30,000	11,094.91	149,781.30		
17		7	149,781.30	-30,000	9,582.50	129,363.81		
18		8	129,363.81	-30,000	7,949.10	107,312.91		
19		9	107,312.91	-30,000	6,185.03	83,497.94		
20		10	83,497.94	-30,000	4,279.84	57,777.78		
21		11	57,777.78	-30,000	2,222.22	30,000.00		
22		12	30,000.00	-30,000	0.00	0.00		

1.7.1 Solving the Retirement Problem Using Financial Formulas

We can develop an even more intelligent solution to the problem if we understand the discounting process. The present value of the whole series of payments, discounted at 8 percent, must be zero.

$$\sum_{t=0}^{4} \frac{\text{Initial deposit}}{(1.08)^t} - \sum_{t=5}^{12} \frac{30{,}000}{(1.08)^t} = 0$$

$$\Rightarrow \text{Initial deposit} = \left[\sum_{t=5}^{12} \frac{30{,}000}{(1.08)^t}\right] \Big/ \left[\sum_{t=0}^{4} \frac{1}{(1.08)^t}\right]$$

Both the numerator on the right-hand side as $\sum_{t=5}^{12} \frac{30{,}000}{(1.08)^t} = \frac{1}{(1.08)^4} \sum_{t=1}^{8} \frac{30{,}000}{(1.08)^t}$ and the denominator $\sum_{t=0}^{4} \frac{1}{(1.08)^t}$ can be calculated using Excel's **PV** function:

	A	B	C	D	E
24	Numerator	126,718.54	<-- =1/(1+B3)^4*PV(B3,8,-30,000)		
25	Denominator	4.31	<-- =PV(B3,5,-1,,1)		
26	Annual deposit	29,386.55	<-- =B24/B25		

[3]Of course you're going to live much longer! And I wish you good health! The dimensions of this problem have been chosen to make it fit nicely on a page.

[4]If the Solver does not appear on the Tools menu, then you have to load it. Go **Tools|Add-Ins** and click **Solver Add-In** on the list of programs. Note that you could also use the **Goal Seek** tool to solve this problem. For simple problems such as this one, there is not much difference between the Solver and Goal Seek; the one (not inconsiderable) advantage of the Solver is that it remembers its previous arguments, so that if you bring it up again on the same spreadsheet, you can see what you did in the previous iteration. In later chapters we will illustrate problems that cannot be solved by Goal Seek and where the use of the Solver is a necessity.



Chapter 1 - Basic Financial Calculations

Financial Modeling, Second Edition

Simon Benninga

1.8 Continuous Compounding

Suppose you deposit $1,000 in a bank account that pays 5 percent per year. At the end of the year you will have 1,000 * (1.05) = $1,050. Now suppose that the bank pays you 2.5 percent interest twice a year. After six months you'll have $1,025, and after one year you will have $1,000 * $\left(1+\frac{0.05}{2}\right)^2$ = $1,050.625. By this logic, if you get paid interest n times per year, your accretion at the end of the year will be $1,000 * $\left(1+\frac{0.05}{n}\right)^n$. As n increases, this amount gets larger, converging (rather quickly, as you will soon see) to $e^{0.05}$, which in Excel is written as the function **Exp.** When n is infinite, we refer to this process as *continuous compounding.*

(By typing **Exp(1)** in a spreadsheet cell, you can see that e = 2.7182818285….)

As you can see in the next display, $1,000 continuously compounded for one year at 5 percent grows to $1,000 * $e^{0.05}$ = $1,051.271 at the end of the year. Continuously compounded for t years, it will grow to $1,000 * $e^{0.05*t}$.



1.8.1 A Technical Note on the Graph

The graph is an Excel XY (Scatter) chart; the x-axis in the chart has been set to be in logarithmic scale. This emphasizes the compounding process. The following picture shows the graph's x-axis marked and the relevant dialog box (right-click after marking the axis and go to **Format Axis**).



1.8.2 Back to Finance—Continuous Discounting

If the accretion factor for continuous compounding at interest r over t years is e^{rt}, then the discount factor for the same period is e^{-rt}. Thus a cash flow C_t occurring in year t and discounted at continuously compounded rate r will be worth $C_t e^{-rt}$ today, as illustrated here.

	B	C	D	E	F	G	H
24	Interest	8%					
25				Continuously			
26		Year	Cash flow	discounted PV			
27		1	100	92.3116	<-- =EXP(-$C$24*C27)*D27		
28		2	200	170.4288			
29		3	300	235.9884			
30		4	350	254.1522			
31		5	400	268.1280			
32							
33		Present value		1021.0089	<-- =SUM(E27:E31)		

1.8.3 Calculating the Continuously Compounded Return from Price Data

Suppose at time 0 you had $1,000 in the bank and suppose that one year later you had $1,200. What was your percentage return? Although the answer may appear obvious, it actually depends on the compounding method. If the bank paid interest only once a year, then the return would be 20 percent:

$$\frac{1{,}200}{1{,}000} - 1 = 20 \text{ percent}$$

However, if the bank paid interest twice a year, you would need to solve the following equation to calculate the return:

$$1{,}000 * \left(1 + \frac{r}{2}\right)^2 = 1{,}200 \Rightarrow r = \left(\frac{1{,}200}{1{,}000}\right)^{1/2} - 1 = 9.5445 \text{ percent}$$

The annual percentage return when interest is paid twice a year is therefore 2 * 9.5445 percent = 19.089 percent.

In general, if there are n compounding periods per year, you have to solve $r = \left(\frac{1{,}200}{1{,}000}\right)^{1/n}$ −1 and then multiply the result appropriately. If n is very large, this solution converges to $r = \ln\left(\frac{1{,}200}{1{,}000}\right)$ = 18.2322 percent:

	A	B	C	D	E	F	G	H	I
37	Initial deposit	1,000					Annual rate with		
38	End-of-year value	1,200					n compounding periods		
39	Number of compounding periods	2							
40	Implied annual interest rate	19.089%	<-- =((B38/B37)^(1/B39)-1)*B39					19.089%	<-- =B40
41							1	20.000%	
42	Continuous return	18.232%	<-- =LN(B38/B37)				2	19.089%	
43							4	18.654%	
44							8	18.442%	
45							20	18.316%	
46							50	18.268%	
47							100	18.249%	

1.8.4 Why Use Continuous Compounding?

All of this may seem somewhat esoteric. However, continuous compounding and discounting are often used in financial calculations. In this book, continuous compounding is used to calculate portfolio returns (Chapters 7–12) and in practically all of the options calculations (Chapters 13–19).

There's another reason to use continuous compounding—its ease of calculation. Suppose, for example, that your \$1,000 grew to \$1,500 in one year and nine months. What's the annualized rate of return? The easiest—and most consistent—way to answer this question is to calculate the continuously compounded annual return. Since one year and nine months equals 1.75 years, this return is

$$1{,}000 * \exp[r * 1.75] = 1{,}500 \Rightarrow r = \frac{1}{1.75}\ln\left[\frac{1{,}500}{1{,}000}\right] = 23.1694 \text{ percent}$$

Exercises

1. You are offered an asset costing \$600 that has cash flows of \$100 at the end of each of the next 10 years.
    a. If the appropriate discount rate for the asset is 8 percent, should you purchase it?
    b. What is the IRR of the asset?
2. You just took a \$10,000, five-year loan. Payments at the end of each year are flat (equal in every year) at an interest rate of 15 percent. Calculate the appropriate loan table, showing the breakdown in each year between principal and interest.
3. You are offered an investment with the following conditions:
    - The cost of the investment is 1,000.
    - The investment pays out a sum *X* at the end of the first year; this payout grows at the rate of 10 percent per year for 11 years.

    If your discount rate is 15 percent, calculate the smallest *X* that would entice you to purchase the asset. For example, as you can see in the following display, *X* = \$100 is too small—the NPV is negative.

	A	B	C	D
2	Discount rate	15%		
3	Initial payment	100		
4	NPV	-226.52		
5				
6	Year			
7	0	-1000.00		
8	1	100.00	<-- =B3	
9	2	110.00	<-- =B8*1.1	
10	3	121.00	<-- =B9*1.1	
11	4	133.10		
12	5	146.41		
13	6	161.05		
14	7	177.16		
15	8	194.87		
16	9	214.36		
17	10	235.79		
18	11	259.37		

4. The following cash-flow pattern has two IRRs. Use Excel to draw a graph of the NPV of these cash flows as a function of the discount rate. Then use the IRR function to identify the two IRRs. Would you invest in this project if the opportunity cost were 20 percent?

	A	B
5	Year	Cash flow
6	0	-500
7	1	600
8	2	300
9	3	300
10	4	200
11	5	-1000

5. In this exercise we solve iteratively for the internal rate of return. Consider an investment that costs 800 and has cash flows of 300, 200, 150, 122, 133 in years 1–5 (see cells A8:B13 in the following spreadsheet). Setting up the loan table shows that 10 percent is greater than the IRR (because the return of principal at the end of year 5 is less than the principal at the beginning of the year).

	A	B	C	D	E	F	G	H	I	J
2				IRR?	10.00%					
3										
4				LOAN TABLE			Division of payment			
5							between interest			
6					Principal	Payment and return of principal				
7	Year			=-B8	at beginning	at end				
8	0	-800		year	of year	of year	Interest	Principal	=F9-G9	
9	1	300		1	800.00	300	80.00	220.00		
10	2	200		2	580.00	200	58.00	142.00		
11	3	150		3	438.00	150	43.80	106.20		
12	4	122		4	331.80	122	33.18	88.82		
13	5	133		5	242.98	133	24.30	108.70		
14										
15				=E9-H9		=$E$2*E9				
16										

Setting the **IRR?** cell equal to 3 percent shows that 3 percent is less than the IRR, since the return of principal at the end of year 5 is greater than the principal at the beginning of year 5.

By changing the **IRR?** cell, find the internal rate of return of the investment.

	D	E	F	G	H	I	J
2	IRR?	3.00%					
3							
4	LOAN TABLE			Division of payment			
5				between interest			
6		Principal	Payment	and return of principal			
7	=-B8	at beginning	at end				
8	year	of year	of year	Interest	Principal	=F9-G9	
9	1	800.00	300	24.00	276.00		
10	2	524.00	200	15.72	184.28		
11	3	339.72	150	10.19	139.81		
12	4	199.91	122	6.00	116.00		
13	5	83.91	133	2.52	130.48		
14	6	-46.57					
15	=E9-H9		=$E$2*E9				

6. An alternative definition of the IRR is the rate that makes the principal at the beginning of year 6 equal to zero. [5] This is shown in the preceding printout, in which cell E14 gives the principal at the beginning of year 6. Using the **Goal Seek** function of Excel, find this rate (we illustrate how the screen should look).



(Of course, you should check your calculations by using the Excel **IRR** function.)

7. Calculate the flat annual payment required to pay off a five-year loan of $100,000 bearing an interest rate of 13 percent.

8. You have just taken a car loan of $15,000. The loan is for 48 months at an annual interest rate of 15 percent (which the bank translates to a monthly rate of 15 percent/12 = 1.25 percent). The 48 payments (to be made at the end of each of the next 48 months) are all equal.
   a. Calculate the monthly payment on the loan.
   b. In a loan table, calculate, for each month, the principal remaining on the loan at the beginning of the month and the split of that month's payment between interest and repayment of principal.
   c. Show that the principal at the beginning of each month is the present value of the remaining loan payments at the loan interest rate (use the **PV** function).

9. You are considering buying a car from a local auto dealer. The dealer offers you one of two payment options:

   - You can pay $30,000 cash.
   - The "deferred payment plan": You can pay the dealer $5,000 cash today and a payment of $1,050 at the end of each of the next 30 months.

   As an alternative to the dealer financing, you have approached a local bank, which is willing to give you a car loan of $25,000 at the rate of 1.25 percent per month.
   a. Assuming that 1.25 percent is the opportunity cost, calculate the present value of all the payments on the dealer's deferred payment plan.
   b. What is the effective interest rate being charged by the dealer? Do this calculation by preparing a spreadsheet like this (only part of the spreadsheet is shown—you have to do this calculation for all 30 months):

	C	D	E	F
14			**Payment under**	
15		**Cash car**	**deferred payment**	
16	**Month**	**payment**	**plan**	**Difference**
17	0	30,000	5,000	25,000
18	1	0	1,050	-1,050
19	2	0	1,050	-1,050
20	3	0	1,050	-1,050
21	4	0	1,050	-1,050
22	5	0	1,050	-1,050
23	6	0	1,050	-1,050

   Now calculate the IRR of the numbers in column F; this is the monthly *effective interest rate* on the deferred payment plan.

10. You are considering a savings plan that calls for a deposit of $15,000 at the end of each of the next five years. If the plan offers an interest rate of 10 percent, how much will you accumulate at the end of year 5?

    Do this calculation by completing the following spreadsheet. This spreadsheet does the calculation twice—once using the **FV** function and once using a simple table that shows the accumulation at the beginning of each year.

	A	B	C	D	E	F	G	H
3	Annual payment	15,000				**Accumulation**	**Payment**	**Interest**
4	Interest rate	10%				**at beginning**	**at end**	**on beg. year**
5	Number of years	5			Year	**of year**	**of year**	**accumulation**
6	Total value	$91,576.50	<-- =FV(B4,B5,-B3,0)		1	0	15,000	0.00
7					2	15,000	15,000	1,500.00
8					3	31,500		
9					4			
10					5			
11					6			

11. Redo the calculation of exercise 10, this time assuming that you make five deposits at the *beginning* of this year and the following four years. How much will you accumulate by the end of year 5?

12. A mutual fund has been advertising that, had you deposited $250 per month in the fund for the last 10 years, you would now have accumulated $85,000. Assuming that these deposits were made at the beginning of each month for a period of 240 months, calculate the effective annual return fund investors got.

    Hint Set up the following spreadsheet, and then use **Goal Seek**

	A	B	C	D
3	Monthly payment	250		
4	Number of months	120		
5				
6	Effective monthly return?			
7	Accumulation		<-- =FV(B6,B4,-B3,,1)	

The effective annual return can then be calculated in one of two ways:

1. $(1 + \text{Monthly return})^{12} - 1$: This is the compound annual return, which is preferable, since it makes allowance for the reinvestment of each month's earnings.
2. 12 * Monthly return: This method is often used by banks.

13. You have just turned 35, and you intend to start saving for your retirement. Once you retire in 30 years (when you turn 65), you would like to have an income of $100,000 per year for the next 20 years. Calculate how much you would have to save between now and age 65 in order to finance your retirement income. Make the following assumptions:

    - All savings draw compound interest of 10 percent per year.
    - You make the first payment today and the last payment on the day you turn 64 (30 payments).
    - You make the first withdrawal when you turn 65 and the last withdrawal when you turn 84 (20 payments).

14. You have $25,000 in the bank, in a savings account that draws 5 percent interest. Your business needs $25,000, and you are considering two options: (a) Use the money in your savings account or (b) borrow the money from the bank at 6 percent, leaving the money in your savings account.

    Your financial analyst suggests that solution (b) is better. His logic: The sum of the interest paid on the 6 percent loan is lower than the interest earned at the same time on the $25,000 deposit. His calculations are illustrated in the following spreadsheet. Show that this logic is wrong. (If you think about it, it couldn't be preferable to take a 6 percent loan when you are getting 5 percent interest from the bank. However, the explanation for this may not be trivial.)

	A	B	C	D	E
1	Financial analysts' calculations				
2					
3	Interest earned	5%			
4	Interest paid	6%			
5	Initial deposit	25,000			
6				=PMT($B$4,2,-$B$5)	
7		THE 6% LOAN			
8	Year	Principal	Payment	Of which	
9		beginning	at end of	Interest	Repayment
10		of year	year	paid	of principal
11	1	25,000.00	13,635.92	1,500.00	12,135.92
12	2	12,864.08	13,635.92	771.84	12,864.08
13		Total interest paid		2,271.84	
14					
15		Savings Account			
16	Year	In savings	End-year	In account	
17		account	interest	at end of	
18		at beginning year	earned	year	
19	1	25,000.00	1,250.00	26,250.00	
20	2	26,250.00	1,312.50	27,562.50	
21		Interest earned	2,562.50		

[5]In general, of course, the IRR is the rate of return that makes the principal in the year *following* the last payment equal to zero.